
Objective
Start structured programming : building a program from named blocks
of code. On this first sheet of 11.3 you learn to define and to call a procedure ,
to give it parameters , and to say where a procedure is appropriate. Sheet 11.3.1
deals with functions, sheet 11.3.2 with the ways of passing parameters, and sheet 11.3.3
with whole programs.

You must be able to:

• define a procedure with PROCEDURE … ENDPROCEDURE, and call it with CALL
• define and call a procedure that has none, one or more parameters
• follow a program that calls procedures, and state what it outputs
• explain where in an algorithm it is appropriate to use a procedure, and give the

benefits

1 Worked examples
Study these first. Each one shows a method that a question later on this sheet needs.

What a procedure is
A procedure is a group of statements with a name. It is defined once and called
(run) from anywhere in the program, as often as it is needed.

PROCEDURE ShowLine()
OUTPUT "--------------------"

ENDPROCEDURE

CALL ShowLine()
OUTPUT "RESULTS"
CALL ShowLine()

• The definition starts with PROCEDURE, the name and brackets, and ends with
ENDPROCEDURE. The first line is the header .

• Defining a procedure does not run it. It runs each time it is called, with CALL

and its name.
• When the procedure ends, the program carries on at the statement after the CALL.

This program outputs a line, then RESULTS, then a line again.

Parameters: giving a procedure its data
A parameter is a variable in the header that receives a value when the procedure
is called. It lets one procedure work on different data.

PROCEDURE Greet(Name : STRING)
OUTPUT "Hello, ", Name

ENDPROCEDURE

CALL Greet("Ada")
CALL Greet(StudentName)

• Each parameter is written as a name, a colon and a data type, like a declaration.
• The value that is given in the call is an argument : "Ada", or the value of

StudentName. It is copied into the parameter Name.
• With two or more parameters, the arguments are matched by position: the first

argument goes into the first parameter, and so on. The number of arguments and
their data types must match the header.

PROCEDURE ShowArea(SideLength : REAL, Width : REAL)
OUTPUT "Area: ", SideLength * Width

ENDPROCEDURE

CALL ShowArea(4.0, 2.5)

Here SideLength receives 4.0 and Width receives 2.5, and the output is Area: 10.0.

Following a program that calls procedures
Method: start at the first statement outside every definition. At each CALL:
note the value of each parameter, run the procedure’s statements, then return to the
statement after the CALL.

PROCEDURE Stars(Number : INTEGER)
DECLARE Count : INTEGER
DECLARE Line : STRING
Line ← ""
FOR Count ← 1 TO Number

Line ← Line & "*"
NEXT Count
OUTPUT Line

ENDPROCEDURE

CALL Stars(3)
CALL Stars(1)

The first call sets Number to 3 and outputs ***; the second sets it to 1 and outputs
*. Count and Line are declared inside the procedure: they are local variables

, which exist only while it runs.

Where a procedure is appropriate

procedure

call Greet(”Ada”)

does an action:
prints ”Hello, Ada”

returns no value

function

set y = Square(5)

computes a value:
5× 5 = 25

returns 25

now y holds 25

Use a procedure for one clear task that gives no value back, above all a task
needed at several points: display a menu, output a report, draw a line of stars. A
task that gives a value back needs a function (sheet 11.3.1).

The benefits of procedures (learn these: they are the marks):

• the statements are written once and called many times, so the program is shorter;
• a change or a correction is made in one place only;
• each procedure can be tested on its own;
• the main program is easier to read, because a meaningful name replaces many

lines;
• different programmers can write different procedures at the same time.

2 Practice
Now use the methods above. The first questions need one idea each.

2.1 Complete the sentences. Use words from the list. Each word is used once. [4]

PROCEDURE Greet(Name : STRING)

CALL Greet("Ada")

argument ·called ·defined ·parameter

A procedure is .................... once, and it can then be .................... many times. A
variable in the header that receives a value is a .................... . The value that the call
gives to it is an .................... .

2.2 Study this pseudocode.

PROCEDURE Welcome()
OUTPUT "Welcome"
OUTPUT "Please log in"

ENDPROCEDURE

OUTPUT "Start"
CALL Welcome()
OUTPUT "End"

program

State everything that is output, in the correct order. [2]

2.3 Define a procedure Greet(Name) that outputs "Hello, " followed by the name. [2]

"Ada" Greet Hello, Ada

2.4 A procedure has the header PROCEDURE ShowPrice(Item : STRING, Price : REAL).

State whether each call is correct. If it is not correct, give the reason. [3]

(a) CALL ShowPrice("Pen", 1.20)
Item : STRING
Price : REAL

ShowPrice

(b) CALL ShowPrice(1.20, "Pen")

(c) CALL ShowPrice("Pen")

2.5 Write a procedure RepeatWord that takes a string Word and an integer Times, and
outputs the word that many times, each on a new line. Then write the call that outputs
"Goal" 3 times. [4]

"Goal"

3
RepeatWord

Goal
Goal
Goal

2.6 State two benefits of using procedures in a program. [2]

PROCEDURE Show()

ENDPROCEDURE

3 Exam-style questions
3.1 Write a procedure Countdown() that takes a positive integer N as a parameter,
outputs the numbers from N down to 1 on separate lines, and then outputs "Go". [5]

3 Countdown

3
2
1
Go

3.2 Study this pseudocode.

PROCEDURE Box(Width : INTEGER, Symbol : CHAR)
DECLARE Count : INTEGER
DECLARE Line : STRING
Line ← ""
FOR Count ← 1 TO Width

Line ← Line & Symbol
NEXT Count
OUTPUT Line

ENDPROCEDURE

CALL Box(4, '#')
CALL Box(2, '+')

(a) State what is output. [2]

CALL Box

CALL Box
PROCEDURE Box screen

(b) Identify the parameters of the procedure, and the arguments of the first call. [2]

(c) State the name of the type of variable that Count and Line are, and state when they
exist. [2]

3.3 A program for a game outputs the same three lines of instructions at the start of the
game, after each level, and when the player asks for help.

(a) Explain why a procedure is appropriate here. [2]

start after a level help

(b) The instructions are changed in a new version of the game. Explain the benefit of
having used a procedure. [2]

3.4 A teacher wants a program that outputs a report for each student: the student’s
name, then a line of text that depends on the mark ("Distinction" for 80 or more,
"Pass" for 50 or more, otherwise "Not yet").

(a) Write a procedure Report that takes a student’s name and mark as parameters, and
outputs the two lines. [6]

"Ada"

85
Report Ada

Distinction

(b) Write a main program that inputs the name and the mark of each of 30 students,
and calls Report for each one. Declare your variables. [5]

Solutions
Each answer shows the steps, then the result.

2.1

• a procedure is defined once and called many times
• the variable in the header is a parameter, and the value given in the call is an

argument

2.2

• the program starts at the first statement outside the definition, so "Start" comes
first

• the call runs both lines of the procedure, and the program then carries on: Start,
Welcome, Please log in, End

2.3

• PROCEDURE … ENDPROCEDURE with a parameter

PROCEDURE Greet(Name : STRING)
OUTPUT "Hello, ", Name

ENDPROCEDURE

2.4

(a) correct: a string and a real number, in the order of the header

(b) not correct: the arguments are in the wrong order, so a number would go into the
STRING parameter Item

(c) not correct: the header has two parameters, but the call gives only one argument

2.5

• a header with the two parameters and their types
• a FOR loop from 1 to Times, with a local counter
• OUTPUT Word inside the loop, and ENDPROCEDURE
• the call: CALL RepeatWord("Goal", 3)

PROCEDURE RepeatWord(Word : STRING, Times : INTEGER)
DECLARE Count : INTEGER
FOR Count ← 1 TO Times

OUTPUT Word
NEXT Count

ENDPROCEDURE

CALL RepeatWord("Goal", 3)

2.6 any two of:

• the statements are written once and called many times, so the program is shorter
• a change is made in one place only

• a procedure can be tested on its own
• the program is easier to read

3.1

• PROCEDURE header with the parameter, and ENDPROCEDURE
• a loop that counts down from N to 1
• output each number inside the loop
• output "Go" once, after the loop

PROCEDURE Countdown(N : INTEGER)
DECLARE Count : INTEGER
FOR Count ← N TO 1 STEP -1

OUTPUT Count
NEXT Count
OUTPUT "Go"

ENDPROCEDURE

• a WHILE loop that decreases a copy of N earns the same marks

3.2

(a) the first call joins four # characters, and the second call joins two + characters

• the output is #### and then ++

(b) the parameters are Width and Symbol

• the arguments of the first call are 4 and '#'

(c) they are local variables

• they exist only while the procedure is running

3.3

(a) the same group of statements is needed at three different points of the algorithm

• the task only outputs text and gives no value back, so a procedure that is written
once and called three times is appropriate

(b) the instructions are written in one place only, inside the procedure

• so one change updates all three places, and no copy can be forgotten or changed
differently

3.4

(a) a header with two parameters and their types

• the name is output
• a first test for 80 or more
• a nested IF for 50 or more, with the ELSE for "Not yet"
• the three messages in the correct paths

• every construct closed, and ENDPROCEDURE

PROCEDURE Report(Name : STRING, Mark : INTEGER)
OUTPUT Name
IF Mark >= 80 THEN

OUTPUT "Distinction"
ELSE

IF Mark >= 50 THEN
OUTPUT "Pass"

ELSE
OUTPUT "Not yet"

ENDIF
ENDIF

ENDPROCEDURE

(b) the variables declared

• a FOR loop that runs 30 times
• the name and the mark input inside the loop
• CALL Report(…) inside the loop
• the arguments in the same order as the parameters

DECLARE Student : INTEGER
DECLARE StudentName : STRING
DECLARE StudentMark : INTEGER
FOR Student ← 1 TO 30

INPUT StudentName
INPUT StudentMark
CALL Report(StudentName, StudentMark)

NEXT Student

